
Zusammenfassung 

Formate kann man als SAS Programmierer nicht wirklich umgehen. Formate liefern tolle 
Funktionalitäten um einfache Lösungen für komplexe Probleme zu erarbeiten. Dieses Pa-
pier zeigt zu dem Thema Gruppieren und Selektieren von Beobachtungen Anwendungsbei-
spiele und Fallen die zu unerwünschten Ergebnissen führen. 

 
Schlüsselwörter: SAS Format, Input, Dezimalzahlen, Selektion, Gruppenbildung 

1 SAS Format: Tricks und Fallen 

1.1 Vorsicht! Format w.d im Input Statement 

Input von Ganzzahlen gemischten mit Dezimalzahlen. Wer bei einem Input Statement 
das Format w.d benutzt kann eine böse Überraschung erleben. Ganzzahlen werden als 
Nachkommastellen eingelesen und führen zu falschen Ergebnissen. 
 
data korb; 

  input kdn_nr name $ richtig falsch 5.2; 

  datalines; 

10 pety 1.00 1.00 

11 carl 0.25 0.25 

12 hans    1    1 

13 mike   41   41 

14 jose  123  123 

15 jimy   5    5  

; 

run; 

 
Schauen wir uns den Output an, dann sehen wir falsche Werte bei der Variablen die das 
w.d Format im Input Statement benutzt. 



S. Rampersad 

276 

proc print data=korb width=uniform noobs; run; 
 

    kdn_nr  name    richtig    falsch 

    10      pety       1.00     1.00  

    11      hans       1.00     0.01  

    12      mike      41.00     0.41  

    13      jose     123.00     1.23  

    14      carl       0.25     0.25  

    15      jim        5.00     0.05  

1.2 Selektion mit einem Format 

Kunden mit einem Format selektieren. Mit einem Format können sehr elegant alle Be-
obachtungen mit einer bestimmten Ausprägung selektiert werden. Wir selektieren mit 
der Variablen IDN aus folgender Datei nur Beobachtungen mit den Werten 2, 4 und 5. 
 

 
 
Zuerst erstellen wir ein Format mit den Werten die selektiert werden sollen.  
proc format; 
    value keyn 
        2,4,5 = 'Y' 
        other = 'N' ; 
run; 
 

Dann benutzen wir das Format um die gewünschten Beobachtungen zu selektieren.  
data subset; 
    set large; 
    where put(idn,keyn.) = 'Y'; 
run; 

Output: 

 

1.3 Mit Längenangaben im Format selektieren 

Mit nur den ersten beiden Stellen wird eine Variablen selektiert. Zuerst erstellen wir das 
passende Format und dann wenden wir es im Where Statement an. 

proc format;                               

 invalue $key  

   'Al','Jo','Ju' = 'Y' 

   other = 'N' ; 

run; 

 

data result; 

 set sashelp.class; 

 where input(name,$key2.) = 'Y'; 

run; 

 

Output: 
proc print width=uniform noobs; 

run; 

 

Input Datensatz  

 

1.4 Format im Format 

Um ein Format zu erweitern ohne es zu ändern integrieren wir ein Format in ein anderes 
Format. Hier wird das Format „sex“ um den Wert „Missing“ erweitert. 

proc format; 

    value sex 

        1='Men' 

        2='Woman' 

        ; 

    value msex 

        .='Missing' 

        other = [sex.] 

        ; 

run; 

Output mit sex Format  
 

Output mit msex Format 

1.5 Gruppenbildung mit einem Format 

Mit den Schlüsselwörtern low und high können alle Werte zugeordnet werden die klei-
ner oder grösser sind als eine bestimme Zahl. 

proc format 

  value grage 

    low-13  = 'jung' 

    14      = 'normal' 

    15-high = 'alt'; 

run; 

 

data view(keep=name age grage); 

  set sashelp.class; 

  format grage grage.; 

  grage=age; 

run; 

 

Output: 
proc print width=uniform noobs; 

run; 

 

 

1.6 Vorsicht! Berechnungen mit Dezimalzahlen 

Computer können nicht rechnen? Für SAS ergibt 15,6 - 14,4 nicht 1,2 sondern 
1,200000000000001. Berechnung mit Dezimalzahlen können von Computern nicht kor-
rekt verarbeitet werden. Ein Ergebnis muss gerundet werden, damit es weiterverarbeitet 
werden kann. Problem ist die begrenzte Möglichkeit Zahlen mit mehreren Stellen zu 
speichern. Ein Format verändert nur die Ausgabe nicht den Wert einer Variablen. 

Input Datensatz: 

data korb; 

  input zahl1 zahl2 subtraktion addition; 

  datalines; 

14.7 11.9 2.8 26.6 

15.4 14.4 1.0 29.8 

; 

run; 

 
Kann SAS wirklich nicht rechnen? Schauen wir uns das Ergebnis genauer an und dann 
wird klar, dass es bei Berechnungen mit Dezimalzahlen zu unerwarteten Ergebnissen 
kommen kann. 

data frucht; 

  set korb; 

  format ergebnis best32. gleich nj.; 

  ergebnis = zahl1 - zahl2;   

  gleich = (ergebnis = subtraktion); 

  output; 

run; 

 

Output: 
zahl1    zahl2    subtraktion          ergebnis    gleich 

 14.7     11.9        2.8      2.79999999999999      nein  

 15.4     14.4        1.0                     1        ja 

 
Kann ein Format das Problem beheben, nein. Ein Format verändert nur die Ausgabe 
nicht den Wert einer Variablen. Das Ergebnis bleibt das gleiche, 2.8 ist nicht gleich 2.8! 
 
data frucht; 

  set korb; 

  format ergebnis best32. ertrag best12. gleich nj.; 

  ergebnis = zahl1 - zahl2; 

  ertrag = ergebnis; 

  gleich = (ertrag = subtraktion); 

 output; 

run; 

 
Output: 
zahl1    zahl2    subtraktion          ergebnis    ertrag   gleich 

 14.7     11.9        2.8      2.79999999999999       2.8     nein  

 15.4     14.4        1.0                     1         1       ja 

Wie kann ich das Ergebnis einer Berechnung mit Dezimalzahlen ohne Probleme ver-
gleichen oder weiterverarbeiten? Der Wert muss gerundet werden. 
 
data frucht; 

  set korb; 

  format ergebnis best32. ertrag best12. gleich nj.; 

  ergebnis = zahl1 - zahl2; 

  ertrag = round(ergebnis,0.1); 

  gleich = (ertrag = subtraktion); 

 output; 

run; 

Output: 
zahl1    zahl2    subtraktion          ergebnis    ertrag   gleich 

 14.7     11.9        2.8      2.79999999999999       2.8       ja 

 15.4     14.4        1.0                     1         1       ja 

1.7 Format mit Informat aus einer Datei erstellen 

Die Reihenfolge mit einem Format ändern. Formate können mehr als nur die Reihenfol-
ge einer Variablen ändern. Ich möchte aber auch dazu einen kleinen Code-Auszug zu 
diesem Thema vorstellen. Die Variable Car soll nach Marke, PS und dann Model sor-
tiert werden. 
 

 
 
Wenn wir alles richtig machen dann werden wir später dieses Ergebnis sehen. Die Vari-
able Car sortiert nach Marke, PS und Model! 

Um eine Variable nach verschieden Kriterien zu sortieren, kann es hilfreich sein, ein 
Format mit passendem Informat zu erstellen. Wir definieren die Länge der Variablen 
und wollen den Formatname vorne als erster Variable stehen haben, dann folgt ein Keep 
Statement. 
 
data fmtout; 

   length fmtname $8 start label help $200 type $1; 

  retain fmtname 'carfmt'; 

  keep fmtname type start label startn; 
  set cars; 

 
Für die richtige Sortierung definieren wir die Variable “startn”. 
 
startn = _n_; /*fuer die sortierung*/ 

 
Jetzt erstelle wir das Format und Informat und definieren die dafür nötigen Variablen 
Type (numerisch oder character), Start und Label. Automatisch wird dann auch die Va-
riable End erzeugt. 
 
/*format*/ 

type = 'N'; 

start = strip(put(_n_, best.)); 

label = strip(car); 

output; 

 
/*informat*/ 

if type eq 'N' then type='I'; 

else if type eq 'C' then type='J'; 

else abort; 

help=strip(start); 

start=strip(label); 

label=strip(help); 

output; 

 
Zum Schluss müssen wir die Formate noch sortieren und als Datei speichern. 
 
/*formate sortieren*/ 

proc sort data=fmtout nodupkey; 

  by fmtname type startn label; 

run; 

/*formate in katalog schreiben*/ 

proc format cntlin=fmtout; 

run; 

Die Variable CARN erstellen wir mit dem Informat und legen dann das Format auf die 
Variable. 
 
format Carn carfmt.; 

Carn=input(car, carfmt.); 

 
Ein Blick auf das Format zeigt uns die Zuordnungen von Start und Label. Die Variable 
CARN ist jetzt nach der gewünschten Reihenfolge sortiert und kann verwendet werden. 

 


